
Abstract

In this paper, we formalized some theorems concerning the product of cyclic groups.

In this article, we present the generalized formalization of [1]. First, we formalize that

every finite commutative group which order is composite number is isomorphic to a

direct product of finite commutative groups which orders are relative prime. Next, we

formalize finite direct products of finite commutative groups.

1 はじめに
有限可換群の基本定理の形式化について報告する．有限可換群の議論は代数学分野で

重要な基礎であり，暗号理論等，情報工学分野でも重要な道具になっている．この報告は，
有限可換群Gには，その位数の素因数分解列に現れる素数因数の累乗を位数とする可換部
分群の族が与えられ，Gはこの族の直積に分解されることの形式化について述べている．

2 有限可換族の直積分解
2.1 形式化の方針
基本的な形式化の方針は，先ず，有限可換群Gの位数 card Gが互いに素な自然数 h, k

の積で表されるとき，Gの部分群でそれぞれの位数が h, k である H,K が存在し，Gが

H,K の直積に分解されることを示す以下の補題 [GROUP 17 - Th.16] の形式化を行い，
ついで，Gの位数 card Gの素因数分解

card G = p1
n1 · p2n2 · · · plnl

によって,この補題を繰り返し適用し，Gが部分群の列

H1,H2, · · ·Hl

の直積に分解することを示すというものである．

Listing 1. GROUP 17 - Th.16

theorem :: GROUP 17:16
for G being finite commutative Group,
h,k be Nat
st card G = h∗k & h,k are coprime holds
ex H,K being strict finite Subgroup of G st
the carrier of H = {x where x is Element of G: x|ˆh = 1 G} &
the carrier of K = {x where x is Element of G: x|ˆk = 1 G} &
H is normal & K is normal
&
(for x be Element of G holds
ex a,b be Element of G st a in H & b in K & x = a∗b)
&
(the carrier of H) /\ (the carrier of K) = {1 G};

ここで，素因数列
p1, p2, · · · , pl

はGの位数 card G によって決まるが，長さ lも，出現する素因数 pi も，位数 card Gに
よって変化し，この列に添え字を付して表すと，今後この定理を利用するたびに，その素
数が何であるかを記述しなくてはならず煩雑化が予想される．そのため，位数 card Gの
素因数分解列中に現れる素数そのものを添え字として使う形式化を試みた．その結果，最
終的に以下の形式化 [GROUP 17 - Th.34] が得られた．

Listing 2. GROUP 17 - Th.34

theorem :: GROUP 17:34
for G being strict finite commutative Group
st card G > 1
holds
ex I be non empty finite set,
F be associative Group−like commutative multMagma−Family of I,
HFG be Homomorphism of product F ,G
st I = support (prime factorization card G )

& (for p be Element of I holds
F.p is strict finite commutative Group
&
F.p is Subgroup of G
&
card (F.p) = (prime factorization card G ).p )

&
(for p,q be Element of I st p <> q holds

(the carrier of (F.p) ) /\ (the carrier of (F.q) ) ={1 G})
&
HFG is bijective
&
for x be (the carrier of G)−valued total I −defined Function
st for p be Element of I holds x.p in F.p
holds x in product F & HFG.x =Product x;

上記は Gの部分群の直積との G同型を与える定理であるが，これを使えば Gの部分
群による直積分解は以下 [GROUP 17 - Th.35] のように得られる．

Listing 3. GROUP 17 - Th.35

theorem :: GROUP 17:35
for G being strict finite commutative Group st card G > 1 holds
ex I be non empty finite set,
F be associative Group−like commutative multMagma−Family of I st
I = support (prime factorization card G)
& (for p be Element of I holds F.p is strict Subgroup of G &
card (F.p) = (prime factorization card G).p) &
(for p,q be Element of I st p <> q holds
(the carrier of (F.p)) /\ (the carrier of (F.q)) = {1 G})
&
(for y be Element of G
ex x be (the carrier of G)−valued total I −defined Function

st (for p be Element of I holds x.p in F.p) & y = Product x)
&
for x1,x2 be (the carrier of G)−valued total I −defined Function st
(for p be Element of I holds x1.p in F.p) &
(for p be Element of I holds x2.p in F.p) &
Product x1 = Product x2 holds x1=x2;

2.2 有限可換群の直積分解
上の記述中に現れる

support (prime factorization card G)

は card Gの素因数全ての集合を表している．

prime factorization card G

は素数全ての集合から自然数への写像で，各素数 p に対して，card Gの素因数分解中に
現れる pの指数を対応させる．ここで，pがその素因数分解中に現れない場合は値 0を対
応させている．f を任意の関数とするとき，supportf は f の台集合を表しており，関数
f の値が 0でない f の定義域の元の全ての集合である．
この I = prime factorization card G によって添え字付けられた Gの元の族や，部

分群の族
{xp}p∈I , {Hp}p∈I

について，それらの積 ∏
p∈I

xp,
∏
p∈I

Hp

を使って形式化を行う．後者については既存のライブラリには以下 [GROUP 7 - Def.2]

で定義されている [2]．

Listing 4. GROUP 7 - Def.2

definition
let I be set, F be multMagma−Family of I;
func product F −> strict multMagma means :: GROUP 7:def 2
the carrier of it = product

Carrier F & for f, g being Element of product Carrier F, i being set st i in I

ex Fi being non empty multMagma, h being Function st Fi = F.i & h = (the multF
of it).(f,g) & h.i = (the multF of Fi).(f.i,g.i);

end;

前者に関しては，既存のライブラリには以下 [GROUP 4 - Def.2]で定義されている [3]．

Listing 5. GROUP 4 - Def.2

definition
let G be non empty multMagma;
let F be FinSequence of the carrier of G;
func Product F −> Element of G equals :: GROUP 4:def 2
(the multF of G) ”∗∗” F;

end;

これは自然数の部分集合

Seg n = {k where k is Nat : 1 ≤ k ∧ k ≤ n}

上の関数である有限列に対しての定義 [2] であり，任意の空でない有限集合上の関数につ
いては定義されていないので，以下 [GROUP 17 - Def.1] のように新たに定義した．

Listing 6. GROUP 17 - Def.1

definition
let G be non empty multMagma,

I be finite set,
b be (the carrier of G)−valued

total I −defined Function;
func Product b −> Element of G means :: GROUP 17:def 1
ex f being FinSequence of the carrier of G

st it = Product f & f = b∗canFS(I);
end;

目的の定理の証明には位数 card Gの素因数分解列の長さによる帰納法を用いる．こ
れに伴い，Gの元の族 {xp}p∈I の積

　
∏
p∈I

xp　

　についてもＩの要素 qを一つ選び,qとそれ以外のＩの要素に分け，∏
p∈I

xp =
∏

p∈I\{q}

xp · xq

のように部分積に分解する必要がある．このため以下の補題を形式化した．

Listing 7. GROUP 17 - Th.8–9

theorem :: GROUP 17:8
for G being commutative Group,
A,B being non empty finite set,
FA be (the carrier of G)−valued total A −defined Function,
FB be (the carrier of G)−valued total B −defined Function,
FAB be (the carrier of G)−valued total A \/ B −defined Function
st A misses B & FAB = FA +∗ FB holds

Product (FAB) = (Product FA) ∗ (Product FB);

theorem :: GROUP 17:9
for G being non empty multMagma,
q be set,
z be Element of G,
f be (the carrier of G)−valued total {q}−defined Function
st f = q .−−> z
holds Product f = z;

冒頭に述べたように，先ず補題 [GROUP 17 - Th.16] の形式化を行った．これは有限
可換群Gの位数 cardGが互いに素な，自然数 h, kの積で表されるとき，Gの部分群でそ
れぞれの位数が h, kであるH,K が存在し，G がH,K の直積に分解されることをを示す
補題である．
このための，一般の群 Gの 2つの正規部分群 A,B について，それらの共通部分が G

の単位元だけからなる場合に A,B の元が可換になること，また A,B の直積から Gへの
同型が与えられることを表す以下の命題を形式化した．

Listing 8. GROUP 17 - Th.11–12

theorem :: GROUP 17:11
for G being Group, A,B being normal Subgroup of G st
(the carrier of A) /\ (the carrier of B) = {1 G} holds
for a,b be Element of G st a in A & b in B holds a∗b = b∗a;

theorem :: GROUP 17:12
for G being Group, A,B being normal Subgroup of G st
(for x be Element of G holds
ex a,b be Element of G st a in A & b in B & x = a∗b)
& (the carrier of A) /\ (the carrier of B) = {1 G} holds
ex h being Homomorphism of product <∗A,B∗>,G st h is bijective
& for a,b be Element of G st a in A & b in B
holds h.(<∗a,b∗>) = a∗b;

以上の一般的な命題の準備をもとに，有限可換群 Gの要素 xで xm = 1G となる全て
の元の集合がGの可換な正規部分群になることを示す補題と，その部分群H の位数と素
因数についての補題を経由して目的の [GROUP 17 - Th.16] を形式化した．

[GROUP 17 - Th.16] からは以下の [GROUP 17 - Th.18]のようにGの直積分解を与
える部分群H,Kの直積とGとの同型の存在定理も直ぐに従う．この定理をGの位数に関
する帰納法を用いて繰り返し適用するために必要な補題 [GROUP 17:2,7] も形式化した。

Listing 9. GROUP 17 - Th.18

theorem :: GROUP 17:18
for G being finite commutative Group,
h,k be non zero Nat
st card G = h∗k & h,k are coprime
ex H,K being strict finite Subgroup of G st
card H = h & card K = k &
(the carrier of H) /\ (the carrier of K) = {1 G} &
ex F being Homomorphism of product <∗H,K∗>,G
st F is bijective
& for a,b be Element of G st a in H & b in K
holds F.(<∗a,b∗>) = a∗b;

2.3 有限可換群の有限直積分解
冒頭に述べたように，Gの位数に関する帰納法は，Gの位数 card Gの素因数分解

card G =
∏
p∈I

pnp

I = prime factorization card G

によって,この補題を繰り返し適用し，Gが部分群の族

{Hp}pinI

の直積に分解することを示すものである．
これには，上の素因数分解の結果を素因数 qとそれ以外の素因数に分け，

card G =
∏

p∈I\{q}

pnp · qnq

のように 2つの積に表し，これらを位数にもつ Gの 2つの部分群 H,K を構成する．
次に，位数 ∏

p∈I\{q}

pnp

をもつ Gの部分群H について，帰納法の仮定によって，その直積分解　

{Hp}p∈\{q}

を構成し，この族に K を加えた G の部分族と G の同型を構成する手順をとった．こ
の手順を実行するために一連の補題 [GROUP 17:19–33] を形式化し，目標とした定理
[GROUP 17 - Th.34] を得た．

3 まとめ
進行中の有限可換群の基本定理の形式化について状況を報告した．形式化数学の視点

からは，有限可換群 Gの位数 card Gの素因数分解中に現れる素数 pを直に Gの直積分
解を与えるGの部分族の添え字に使ったところに従来にない形式化の特徴がある．なお，
直積分解に現れた部分群 Hp は可換 p-群な部分群になる．素数 pの累乗を位数とする群
は p-群と呼ばれるが，可換な p-群はいくつかの巡回群の族の直積と同型になることが知
られている．
以後の形式化でこの定理を引用する際の表現の簡潔化が期待できる．有限群の議論は

周知のように未解決課題もあり，精緻で深い内容を持っている．本稿の読者にはこれに興
味を持たれ，筆者らへの助力やライブラリ構築に参加して頂ければ幸いである．

参考文献
[1] Arai K, Okazaki H, Shidama Y. Isomorphisms of Direct Products of Finite Cyclic

Groups. Formalized Mathematics. 2012;20(4):343–347.

MMA-WiP (2021), vol.3, no.0 - https://mizar-jp.org/journal/MMA-WiP/j2021.3.0 - ISSN 2434-5458 (Online) 6/12

https://mizar-jp.org/journal/MMA-WiP/j2021.3.0


Mechanized Mathematics and Its Applications, Works in Progress 岡崎 他, 有限可換群の基本定理

[2] Korni lowicz A. The Product of the Families of the Groups. Formalized Mathemat-

ics. 1998;7(1):127–134. Available from: http://fm.mizar.org/1998-7/pdf7-1/

group_7.pdf.

[3] Trybulec WA. Lattice of Subgroups of a Group. Frattini Subgroup. Formalized

Mathematics. 1991;2(1):41–47. Available from: http://fm.mizar.org/1991-2/

pdf2-1/group_4.pdf.

Mizar article information

Works in Progress

GROUP 17 Isomorphisms of Direct Products of Finite Commutative Groups

by Hiroyuki Okazaki, Hiroshi Yamazaki and Yasunari Shidama

Summary: We have been working on the formalization of groups. In [1], we encoded

some theorems concerning the product of cyclic groups. In this article, we present

the generalized formalization of [1]. First, we show that every finite commutative group

which order is composite number is isomorphic to a direct product of finite commutative

groups which orders are relatively prime. Next, we describe finite direct products of

finite commutative groups.

Listing 10. GROUP 17 - abstract

:: Isomorphisms of Direct Products of Finite Commutative Groups
:: by Hiroyuki Okazaki , Hiroshi Yamazaki and Yasunari Shidama

environ

vocabularies FINSEQ 1, FUNCT 1, RELAT 1, RLVECT 2, CARD 3, TARSKI, BINOP 1,
GROUP 1, XXREAL 0, GROUP 2, CARD 1, FUNCT 4, GROUP 6, GROUP 7, FUNCOP 1,
ALGSTR 0, PARTFUN1, FUNCT 2, SUBSET 1, XBOOLE 0, STRUCT 0, NAT 1,
ORDINAL4, PRE TOPC, ARYTM 1, ARYTM 3, FINSET 1, INT 2, ZFMISC 1, PBOOLE,
NEWTON, INT 1, NAT 3, REAL 1, PRE POLY, XCMPLX 0, UPROOTS, INT 7;

notations TARSKI, XBOOLE 0, ZFMISC 1, SUBSET 1, RELAT 1, FUNCT 1, ORDINAL1,
RELSET 1, PARTFUN1, FUNCT 2, DOMAIN 1, FUNCOP 1, FUNCT 4, FINSET 1,
CARD 1, PBOOLE, CARD 3, NUMBERS, XCMPLX 0, XXREAL 0, XREAL 0, NAT 1,
INT 1, INT 2, BINOP 1, FINSEQ 1, NEWTON, PRE POLY, NAT 3, STRUCT 0,
ALGSTR 0, GROUP 1, GROUP 2, GROUP 3, GROUP 4, GROUP 6, PRALG 1, GROUP 7,
INT 7;

constructors BINOP 1, REALSET1, GROUP 6, MONOID 0, PRALG 1, GROUP 4, CARD 2,
GROUP 7, RELSET 1, WELLORD2, NAT D, INT 7, RECDEF 1, NAT 3, FINSOP 1;

registrations XBOOLE 0, XREAL 0, STRUCT 0, GROUP 2, MONOID 0, FUNCT 2, CARD 1,
CARD 3, GROUP 7, GROUP 3, RELSET 1, FINSEQ 1, INT 1, AOFA 000, GR CY 1,
FINSET 1, NAT 3, RELAT 1, FUNCT 1, MEMBERED, FUNCOP 1, NEWTON, VALUED 0,
PRE POLY, PBOOLE, INT 7, GROUP 6, ORDINAL1;

requirements NUMERALS, SUBSET, ARITHM, BOOLE;

begin :: Preliminaries

theorem :: GROUP 17:1
for A,B,A1,B1 be set st A misses B
& A1 c= A & B1 c= B & A1 \/ B1 = A \/ B holds
A1 = A & B1 = B;

theorem :: GROUP 17:2
for H,K be non empty finite set holds
card product (<∗ H, K ∗>) = card(H)∗card(K);

theorem :: GROUP 17:3
for ps,pt,f be bag of SetPrimes,
q being Nat
st (support ps) misses (support pt) & f = ps + pt & q in (support ps) holds
ps.q = f.q;

theorem :: GROUP 17:4
for ps,pt,f be bag of SetPrimes,
q being Nat
st (support ps) misses (support pt) & f = ps + pt & q in (support pt) holds
pt.q = f.q;

theorem :: GROUP 17:5
for h be non zero Nat, q being Prime
st not q,h are coprime holds
q divides h;

theorem :: GROUP 17:6
for h,s be non zero Nat
st for q being Prime st q in support (prime factorization s)
holds not q,h are coprime holds
support (prime factorization s) c= support (prime factorization h);

theorem :: GROUP 17:7
for h,k,s,t be non zero Nat
st h,k are coprime & s ∗ t = h ∗ k
& (for q being Prime st q in support prime factorization s
holds not q,h are coprime)
& (for q being Prime st q in support prime factorization t
holds not q,k are coprime)
holds
s = h & t = k;

definition
let G be non empty multMagma,
I be finite set,
b be (the carrier of G)−valued total I −defined Function;
func Product b −> Element of G means

:: GROUP 17:def 1
ex f being FinSequence of G st it = Product f & f = b∗canFS(I);

end;

theorem :: GROUP 17:8
for G being commutative Group,
A,B being non empty finite set,
FA be (the carrier of G)−valued total A −defined Function,
FB be (the carrier of G)−valued total B −defined Function,
FAB be (the carrier of G)−valued total A \/ B −defined Function
st A misses B & FAB = FA +∗ FB holds
Product (FAB) = (Product FA) ∗ (Product FB);

theorem :: GROUP 17:9
for G being non empty multMagma,
q be set,
z be Element of G,
f be (the carrier of G)−valued total {q}−defined Function
st f = q .−−> z
holds Product f = z;

begin :: Direct Product of Finite Commutative Groups

theorem :: GROUP 17:10
for X,Y being non empty multMagma holds
the carrier of product <∗X,Y∗>
= product <∗ the carrier of X,the carrier of Y ∗>;

theorem :: GROUP 17:11
for G being Group, A,B being normal Subgroup of G st

(the carrier of A) /\ (the carrier of B) = {1 G} holds
for a,b be Element of G st a in A & b in B holds a∗b = b∗a;

theorem :: GROUP 17:12
for G being Group, A,B being normal Subgroup of G st
(for x be Element of G holds
ex a,b be Element of G st a in A & b in B & x = a∗b)
& (the carrier of A) /\ (the carrier of B) = {1 G} holds
ex h being Homomorphism of product <∗A,B∗>,G st h is bijective
& for a,b be Element of G st a in A & b in B
holds h.(<∗a,b∗>) = a∗b;

theorem :: GROUP 17:13
for G being finite commutative Group,
m be Nat,
A be Subset of G
st A ={x where x is Element of G: x|ˆm = 1 G}
holds
A <> {}
&
(for g1,g2 be Element of G
st g1 in A & g2 in A holds g1 ∗ g2 in A) &
for g be Element of G st g in A holds g” in A;

theorem :: GROUP 17:14
for G being finite commutative Group,
m be Nat,
A be Subset of G
st A ={x where x is Element of G: x|ˆm = 1 G} holds
ex H being strict finite Subgroup of G
st the carrier of H = A & H is commutative normal;

theorem :: GROUP 17:15
for G being finite commutative Group,
m be Nat,
H being finite Subgroup of G
st the carrier of H = {x where x is Element of G: x|ˆm = 1 G} holds
for q being Prime st q in support prime factorization card H
holds not q,m are coprime;

theorem :: GROUP 17:16
for G being finite commutative Group,
h,k be Nat
st card G = h∗k & h,k are coprime holds
ex H,K being strict finite Subgroup of G st
the carrier of H = {x where x is Element of G: x|ˆh = 1 G} &
the carrier of K = {x where x is Element of G: x|ˆk = 1 G} &
H is normal & K is normal
&
(for x be Element of G holds
ex a,b be Element of G st a in H & b in K & x = a∗b)
&
(the carrier of H) /\ (the carrier of K) = {1 G};

theorem :: GROUP 17:17
for H,K be finite Group holds
card product (<∗ H, K ∗>) = card(H)∗card(K);

theorem :: GROUP 17:18
for G being finite commutative Group,
h,k be non zero Nat
st card G = h∗k & h,k are coprime
ex H,K being strict finite Subgroup of G st
card H = h & card K = k &
(the carrier of H) /\ (the carrier of K) = {1 G} &
ex F being Homomorphism of product <∗H,K∗>,G
st F is bijective
& for a,b be Element of G st a in H & b in K
holds F.(<∗a,b∗>) = a∗b;

begin :: Finite Direct Products of Finite Commutative Groups

theorem :: GROUP 17:19
for G be Group,
q be set,
F be associative Group−like multMagma−Family of {q},
f being Function of G,product F st F = q .−−> G &
for x being Element of G holds f . x = q .−−> x holds
f is Homomorphism of G,(product F);

theorem :: GROUP 17:20
for G be Group,
q be set,
F be associative Group−like multMagma−Family of {q},
f being Function of G,product F st F = q .−−> G &
for x being Element of G holds f . x = q .−−> x holds
f is bijective;

theorem :: GROUP 17:21
for q be set,
F be associative Group−like multMagma−Family of {q},
G be Group st F = q .−−> G holds
ex I be Homomorphism of G,product F st
I is bijective &
for x being Element of G holds I . x = q .−−> x;

theorem :: GROUP 17:22
for I0,I be non empty finite set,
F0 be associative Group−like multMagma−Family of I0,
F be associative Group−like multMagma−Family of I,
H,K be Group,
q be Element of I,
k be Element of K,
g be Function st
g in the carrier of product F0 &
not q in I0 & I = I0 \/ {q} & F = F0 +∗ (q .−−> K) holds
g +∗ (q .−−> k) in the carrier of product F;

theorem :: GROUP 17:23
for I0,I be non empty finite set,
F0 be associative Group−like multMagma−Family of I0,
F be associative Group−like multMagma−Family of I,
H,K be Group,
q be Element of I,
G0 be Function of H,product F0 st
G0 is Homomorphism of H,product F0
& G0 is bijective & not q in I0 & I = I0 \/ {q} & F = F0 +∗ (q .−−> K) holds
for G be Function of product <∗H,K∗>,(product F) st
for h be Element of H,k be Element of K
holds ex g be Function
st g=G0.h & G.(<∗h,k∗>) = g +∗ (q .−−> k) holds
G is Homomorphism of product <∗H,K∗>,product F;

theorem :: GROUP 17:24
for I0,I be non empty finite set,
F0 be associative Group−like multMagma−Family of I0,
F be associative Group−like multMagma−Family of I,
H,K be Group,
q be Element of I,
G0 be Function of H, product F0 st
G0 is Homomorphism of H, product F0
& G0 is bijective
& not q in I0 & I = I0 \/ {q} & F = F0 +∗ (q .−−> K) holds
for G be Function of product <∗H,K∗>, product F st
for h be Element of H,k be Element of K
holds ex g be Function
st g=G0.h & G.(<∗h,k∗>) = g +∗ (q .−−> k)
holds G is bijective;

theorem :: GROUP 17:25
for q be set,
F be multMagma−Family of {q},
G be non empty multMagma st
F = q .−−> G holds
for y be (the carrier of G)−valued total {q} −defined Function holds
y in the carrier of product F & y.q in the carrier of G &
y= q .−−> y.q;

theorem :: GROUP 17:26
for q be set,
F be associative Group−like multMagma−Family of {q},
G be Group st F = q .−−> G holds
ex HFG be Homomorphism of product F,G st
HFG is bijective &
for x be (the carrier of G)−valued total {q} −defined Function
holds HFG.x = Product x;

theorem :: GROUP 17:27
for I0,I be non empty finite set,
F0 be associative Group−like multMagma−Family of I0,
F be associative Group−like multMagma−Family of I,
H,K be Group,
q be Element of I,
G0 be Homomorphism of H,(product F0) st
not q in I0 & I = I0 \/ {q} & F = F0 +∗ (q .−−> K) & G0 is bijective
ex G be Homomorphism of product <∗H,K∗>,(product F) st
G is bijective &
for h be Element of H,k be Element of K
ex g be Function st g=G0.h & G.(<∗h,k∗>) = g +∗ (q .−−> k);

theorem :: GROUP 17:28
for I0,I be non empty finite set,
F0 be associative Group−like multMagma−Family of I0,
F be associative Group−like multMagma−Family of I,
H,K be Group,
q be Element of I,
G0 be Homomorphism of product F0, H st not q in I0 &
I = I0 \/ {q} & F = F0 +∗ (q .−−> K) & G0 is bijective holds
ex G be Homomorphism of product F, product <∗H,K∗> st G is bijective &
for x0 be Function,
k be Element of K,
h be Element of H
st h = G0.x0 & x0 in product F0 holds
G.(x0 +∗ (q .−−>k)) = <∗ h, k ∗>;

theorem :: GROUP 17:29
for I be non empty finite set,
F be associative Group−like multMagma−Family of I,
x be total I −defined Function
st for p be Element of I holds x.p in F.p
holds x in the carrier of product F;

theorem :: GROUP 17:30
for I0,I be non empty finite set,
F0 be associative Group−like multMagma−Family of I0,
F be associative Group−like multMagma−Family of I,
K be Group,
q be Element of I,
x be Element of product F st
not q in I0 & I = I0 \/ {q} & F = F0 +∗ (q .−−> K) holds
ex x0 be total I0 −defined Function,
k be Element of K st x0 in product F0
& x = x0 +∗ (q .−−> k) & for p be Element of I0 holds x0.p in F0.p;

theorem :: GROUP 17:31
for G be Group,
H be Subgroup of G,

MMA-WiP (2021), vol.3, no.0 - https://mizar-jp.org/journal/MMA-WiP/j2021.3.0 - ISSN 2434-5458 (Online) 11/12

https://mizar-jp.org/journal/MMA-WiP/j2021.3.0


Mechanized Mathematics and Its Applications, Works in Progress 岡崎 他, 有限可換群の基本定理

f being FinSequence of G,
g being FinSequence of H
st f=g
holds Product f = Product g;

theorem :: GROUP 17:32
for I be non empty finite set,
G be Group,
H be Subgroup of G,
x be (the carrier of G)−valued total I −defined Function,
x0 be (the carrier of H)−valued total I −defined Function
st x=x0
holds Product x = Product x0;

theorem :: GROUP 17:33
for G being commutative Group,
I0,I be non empty finite set,
q be Element of I,
x be (the carrier of G)−valued total I −defined Function,
x0 be (the carrier of G)−valued total I0 −defined Function,
k be Element of G st
not q in I0 & I = I0 \/ {q} & x = x0 +∗ (q .−−> k)
holds
Product x = (Product x0)∗k;

theorem :: GROUP 17:34
for G being strict finite commutative Group
st card G > 1 holds
ex I be non empty finite set,
F be associative Group−like commutative multMagma−Family of I,
HFG be Homomorphism of product F,G st
I = support (prime factorization card G)
& (for p be Element of I holds F.p is strict Subgroup of G &
card (F.p) = (prime factorization card G).p) &
(for p,q be Element of I st p <> q holds
(the carrier of (F.p)) /\ (the carrier of (F.q)) = {1 G}) &
HFG is bijective &
for x be (the carrier of G)−valued total I −defined Function
st for p be Element of I holds x.p in F.p
holds x in product F & HFG.x = Product x;

theorem :: GROUP 17:35
for G being strict finite commutative Group st card G > 1 holds
ex I be non empty finite set,
F be associative Group−like commutative multMagma−Family of I st
I = support (prime factorization card G)
& (for p be Element of I holds F.p is strict Subgroup of G &
card (F.p) = (prime factorization card G).p) &
(for p,q be Element of I st p <> q holds
(the carrier of (F.p)) /\ (the carrier of (F.q)) = {1 G})
&
(for y be Element of G
ex x be (the carrier of G)−valued total I −defined Function

st (for p be Element of I holds x.p in F.p) & y = Product x)
&
for x1,x2 be (the carrier of G)−valued total I −defined Function st
(for p be Element of I holds x1.p in F.p) &
(for p be Element of I holds x2.p in F.p) &
Product x1 = Product x2 holds x1=x2;

